
F08 – Least Squares and Eigenvalue Problems (LAPACK)

F08GUF (CUPMTR/ZUPMTR) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F08GUF (CUPMTR/ZUPMTR) multiplies an arbitrary complex matrixC by the complex unitary matrix
Q which was determined by F08GSF (CHPTRD/ZHPTRD) when reducing a complex Hermitian matrix
to tridiagonal form.

2 Specification

SUBROUTINE F08GUF(SIDE, UPLO, TRANS, M, N, AP, TAU, C, LDC, WORK,
1 INFO)
ENTRY cupmtr(SIDE, UPLO, TRANS, M, N, AP, TAU, C, LDC, WORK,
1 INFO)
INTEGER M, N, LDC, INFO
complex AP(∗), TAU(∗), C(LDC,∗), WORK(∗)
CHARACTER∗1 SIDE, UPLO, TRANS

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine is intended to be used after a call to F08GSF (CHPTRD/ZHPTRD), which reduces a
complex Hermitian matrix A to real symmetric tridiagonal form T by a unitary similarity transformation:
A = QTQH. F08GSF represents the unitary matrix Q as a product of elementary reflectors.

This routine may be used to form one of the matrix products

QC, QHC, CQ or CQH ,

overwriting the result on C (which may be any complex rectangular matrix).

A common application of this routine is to transform a matrix Z of eigenvectors of T to the matrix QZ
of eigenvectors of A.

4 References

[1] Golub G H and van Loan C F (1996) Matrix Computations Johns Hopkins University Press (3rd
Edition), Baltimore

5 Parameters

1: SIDE — CHARACTER*1 Input

On entry: indicates how Q or QH is to be applied to C as follows:

if SIDE = ’L’, then Q or QH is applied to C from the left;
if SIDE = ’R’, then Q or QH is applied to C from the right.

Constraint: SIDE = ’L’ or ’R’.

2: UPLO — CHARACTER*1 Input

On entry: this must be the same parameter UPLO as supplied to F08GSF (CHPTRD/ZHPTRD).

Constraint: UPLO = ’U’ or ’L’.

3: TRANS — CHARACTER*1 Input

On entry: indicates whether Q or QH is to be applied to C as follows:

if TRANS = ’N’, then Q is applied to C;
if TRANS = ’C’, then QH is applied to C.

Constraint: TRANS = ’N’ or ’C’.

4: M — INTEGER Input

On entry: m, the number of rows of the matrix C; m is also the order of Q if SIDE = ’L’.

Constraint: M ≥ 0.

5: N — INTEGER Input

On entry: n, the number of columns of the matrix C; n is also the order of Q if SIDE = ’R’.

Constraint: N ≥ 0.

6: AP(∗) — complex array Input

Note: the dimension of the array AP must be at least max(1,M∗(M+1)/2) if SIDE = ’L’ and at
least max(1,N∗(N+1)/2) if SIDE = ’R’.

On entry: details of the vectors which define the elementary reflectors, as returned by F08GSF
(CHPTRD/ZHPTRD).

7: TAU(∗) — complex array Input

Note: the dimension of the array TAU must be at least max(1,M−1) if SIDE = ’L’ and at least
max(1,N−1) if SIDE = ’R’.

On entry: further details of the elementary reflectors, as returned by F08GSF
(CHPTRD/ZHPTRD).

8: C(LDC,∗) — complex array Input/Output

Note: the second dimension of the array C must be at least max(1,N).

On entry: the m by n matrix C.

On exit: C is overwritten by QC or QHC or CQH or CQ as specified by SIDE and TRANS.

9: LDC — INTEGER Input

On entry: the first dimension of the array C as declared in the (sub)program from which F08GUF
(CUPMTR/ZUPMTR) is called.

Constraint: LDC ≥ max(1,M).

10: WORK(∗) — complex array Workspace

Note: the dimension of the array WORK must be at least max(1,N) if SIDE = ’L’ and at least
max(1,M) if SIDE = ’R’.

11: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

7 Accuracy

The computed result differs from the exact result by a matrix E such that

‖ E ‖2 = O(ε)‖ C ‖2,

where ε is the machine precision.

8 Further Comments

The total number of real floating-point operations is approximately 8m2n if SIDE = ’L’ and 8mn2 if
SIDE = ’R’.

The real analogue of this routine is F08GGF (SOPMTR/DOPMTR).

9 Example

To compute the two smallest eigenvalues, and the associated eigenvectors, of the matrix A, where

A =



−2.28 + 0.00i 1.78− 2.03i 2.26 + 0.10i −0.12 + 2.53i
1.78 + 2.03i −1.12 + 0.00i 0.01 + 0.43i −1.07 + 0.86i
2.26− 0.10i 0.01− 0.43i −0.37 + 0.00i 2.31− 0.92i

−0.12− 2.53i −1.07− 0.86i 2.31 + 0.92i −0.73 + 0.00i


 ,

using packed storage. Here A is Hermitian and must first be reduced to tridiagonal form T by F08GSF
(CHPTRD/ZHPTRD). The program then calls F08JJF (SSTEBZ/DSTEBZ) to compute the requested
eigenvalues and F08JXF (CSTEIN/ZSTEIN) to compute the associated eigenvectors of T . Finally
F08GUF (CUPMTR/ZUPMTR) is called to transform the eigenvectors to those of A.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F08GUF Example Program Text
* Mark 16 Release. NAG Copyright 1992.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, LDZ
PARAMETER (NMAX=8,LDZ=NMAX)
real ZERO
PARAMETER (ZERO=0.0e0)

* .. Local Scalars ..
real VL, VU
INTEGER I, IFAIL, INFO, J, M, N, NSPLIT
CHARACTER UPLO

* .. Local Arrays ..
complex AP(NMAX*(NMAX+1)/2), TAU(NMAX), WORK(NMAX),

+ Z(LDZ,NMAX)
real D(NMAX), E(NMAX), RWORK(5*NMAX), W(NMAX)
INTEGER IBLOCK(NMAX), IFAILV(NMAX), ISPLIT(NMAX),

+ IWORK(NMAX)
CHARACTER CLABS(1), RLABS(1)

* .. External Subroutines ..
EXTERNAL sstebz, X04DBF, chptrd, cstein, cupmtr

* .. Executable Statements ..
WRITE (NOUT,*) ’F08GUF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
* Read A from data file
*

READ (NIN,*) UPLO
IF (UPLO.EQ.’U’) THEN

READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
ELSE IF (UPLO.EQ.’L’) THEN

READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
END IF

*
* Reduce A to tridiagonal form T = (Q**H)*A*Q
*

CALL chptrd(UPLO,N,AP,D,E,TAU,INFO)
*
* Calculate the two smallest eigenvalues of T (same as A)
*

CALL sstebz(’I’,’B’,N,VL,VU,1,2,ZERO,D,E,M,NSPLIT,W,IBLOCK,
+ ISPLIT,RWORK,IWORK,INFO)

*
WRITE (NOUT,*)
IF (INFO.GT.0) THEN

WRITE (NOUT,*) ’Failure to converge.’
ELSE

WRITE (NOUT,*) ’Eigenvalues’
WRITE (NOUT,99999) (W(I),I=1,M)

*
* Calculate the eigenvectors of T, storing the result in Z
*

CALL cstein(N,D,E,M,W,IBLOCK,ISPLIT,Z,LDZ,RWORK,IWORK,
+ IFAILV,INFO)

*
IF (INFO.GT.0) THEN

WRITE (NOUT,*) ’Failure to converge.’
ELSE

*
* Calculate the eigenvectors of A = Q * (eigenvectors of T)
*

CALL cupmtr(’Left’,UPLO,’No transpose’,N,M,AP,TAU,Z,LDZ,
+ WORK,INFO)

*
* Print eigenvectors
*

WRITE (NOUT,*)
IFAIL = 0

*
CALL X04DBF(’General’,’ ’,N,M,Z,LDZ,’Bracketed’,’F7.4’,

+ ’Eigenvectors’,’Integer’,RLABS,’Integer’,
+ CLABS,80,0,IFAIL)

*
END IF

END IF
END IF
STOP

*

99999 FORMAT (8X,4(F7.4,11X,:))
END

9.2 Program Data

F08GUF Example Program Data
4 :Value of N
’L’ :Value of UPLO

(-2.28, 0.00)
( 1.78, 2.03) (-1.12, 0.00)
( 2.26,-0.10) ( 0.01,-0.43) (-0.37, 0.00)
(-0.12,-2.53) (-1.07,-0.86) ( 2.31, 0.92) (-0.73, 0.00) :End of matrix A

9.3 Program Results

F08GUF Example Program Results

Eigenvalues
-6.0002 -3.0030

Eigenvectors
1 2

1 ( 0.7299, 0.0000) (-0.2595, 0.0000)
2 (-0.1663,-0.2061) ( 0.5969, 0.4214)
3 (-0.4165,-0.1417) (-0.2965,-0.1507)
4 ( 0.1743, 0.4162) ( 0.3482, 0.4085)

